DECUS
PROGRAM LIBRARY

DECUS NO.	8-150
TITLE	PTOD8 HIGH AND PTOD8 LOW
AUTHOR	R.A. Gruenewald
COMPANY	Dr. Neher Laboratory Netherlands Postal and Telecommunication Service Leidschendam, The Netherlands
DATE	May 29, 1968
SOURCE LANGUAGE	

PTOD8 - PTT TRACE AND OCTAL ON-LINE DEBUGGING PROGRAM

DECUS Program Library Write-up DECUS No. 8-150

Abstract

PTOD8 (PTT Trace and Octal On-Line Debugging Program for PDP-8) is a program to debug a running program.

Communication between operator and program is via the ASR-33 Teletype.

Features

- Register examination and modification
- Multiple breakpoints (break-traps)
- Memory protection (of a chosen block)
- Word search (masked or not masked)
- Tracing a running program (gives a full print out of consequently executed instructions)
- It is interrupt proof (ION in investigated program is dealt with to make a genuine interrupt)
- Binary tape punching (automatic leader-trailer code and checksum).

Requirements

Storage

PTOD8 requires 1343 (octal) registers.
PTOD8 HIGH is stored between 6200 and 7543.
PTOD8 LOW is stored between 200 and 1543.

Equipment

Standard PDP-8

Usage

Loading

PTOD8 is loaded by means of the Binary Loader (8-2-U).

Place PTOD8 (High or Low) in proper reader (ASR-33 or 750).

Switch register to 7777 and press Load Address.

For type 750 reader set switch 0 down and press START.

For ASR-33 put switch 0 up and press START.

For PTOD8 version B press CONTINUE switch.

Load binary tape of program to be debugged.

Start up.

PTOD8 has starting address 200 (LOW) and 6200 (HIGH)

Start PTOD8 with switch register and control switches.
PTOD8 is now running and awaiting commands from ASR-33 Teletype.

Restrictions

Status Active Registers

Register 0001 and 0002 are used by PTOD8 when Program to be debugged involves ION. In case an ION is executed (by means of PTOD8) a trap is placed in locations to 0001 and 0002 as follows:

0001, JMP I 2

0002, ADDRESS OF INTERRUPT DEALING ROUTINE OF PTOD8.

As soon as genuine interrupt occurs the interrupt dealing routine restores locations 0001 and 0002 with the proper instructions of the program to be debugged.

By thus interchanging traps and proper instructions it is taken care of that program to be debugged may use registers 0001 and 0002 unaware of the fact that PTOD8 uses the same locations.

This only holds for programs with ION in them. When there is no ION, PTOD8 only uses locations 200-1543 (LOW) or 6200-7543 (HIGH).

Status Core

PTOD8 will not operate outside of the core memory module in which it resides.

Description

PTOD8 is an octal debugging program for PDP-8. Communication between operator and PTOD8 occurs on ASR-33 Teletype.

Commands

The register examination character / causes the contents of the register addressed by the preceding expression to be typed out in octal. The register is then opened for modification. (In all examples everything between double quotes (") is typed by the operator and everything underlined is typed by PTOD8.)

Carriage Return (cr)	The value of any expression that has been typed by the user immediately preceding the carriage return will replace the contents of the open register (if opened anyway) and the register is closed. If nothing has been typed, the contents will not be changed.
Line Feed (lf)	Line feed has the same effect as carriage return, but, in addition, the next sequential register is opened and its contents typed.
Correction ←	Any typed octal number can be neglected by typing ←—.
Semicolon (;)	Separation of first address and last address of a chosen block in memory. Used for memory protection, word search and punch binary tape.
Asterisk (*)	Asterisk is not printed. Used for checksum and trailer punch. Trailer is automatically punched after accumulated checksum has been punched. So, ASR-33 can be left on line.
Ampersand (&)	The octal number that has been typed by the user immediately preceding the & will set the MASK used for Word Search. Initially this MASK is set to 7777. After each Word Search the Mask is reset to 7777.
W (Word Search Block)	The number typed by the user preceding W sets the upper limit of the block in memory used for Word Search.
Z (Search)	PTOD8 will perform a Word Search and print the address and contents of every register in the desired block of memory whose contents, when mashed, are equal to the value of the number typed by the user preceding Z.

A (Accu)	Typing the letter A causes the contents of the Accumulator to be printed and opened for modification. A is the accumulator as it occurs in the running program under investigation.
L (Link)	Typing the letter L causes the contents of the LINK to be printed and opened for modification. L is the LINK as it occurs in the running program under investigation.
B (Breakpoint)	Insert a breakpoint at the address printed by the user immediately preceding command B. A maximum of 40 (octal) breakpoints can be inserted at one time.
R (Remove)	Remove the breakpoint that was previously inserted at the address given by the number immediately preceding R.
E (Erase)	Remove all previously inserted breakpoints.
T (Trace)	The number typed by the user immediately preceding T gives the address where PTOD8 will start to trace (print out consequently executed instructions) the program to be debugged. Cancelling the T-mode can be done by: "mT" where m> the highest reachable address of the program to be debugged. Restart PTOD8.
G (Go)	Start the program to be debugged at address typed by the user immediately preceding G.
C (Continue)	When PTOD8 has returned command to the user (in a breakpoint or in trace mode, etc.) typing letter C means to continue executing instructions.
S (Skip)	When PTOD8 has returned command to the user (in a breakpoint or in a trace mode) typing letter S means to execute an unconditional Skip.

	This can be useful when the instruction to be executed is, for example, an ISZ or when the program to be debugged has a reading sequence.
M (Memory Protection)	The number typed by the user immediately preceding M is the upper boundary of the section in memory that is protected. Whenever the running program uses any address within this section of memory PTOD8 will print a message.
P (Punch Binary Tape)	Number typed by the user preceding letter B is upper limit of a block of instruction to be punched in binary format. After command P the PTOD8 provides leader code automatically provided the previous command of the user was an Asterisk (* for checksum and trailer code). It is thus possible to punch consequent block in binary format whereas only the first time leader code will be punched.
Illegal Character	PTOD8 punches ? and returns command.

Switch Register

When PTOD8 is dealing with an object program in Trace Mode, switch register setting means:

SR = 0	Print every executed instruction of object program without returning command to user.
SR ≠ 0	Print one executed instruction of object program and return command to user.
SR(0) = 1	(PTOD8 version B only.) Print next instruction to be executed and return command to user. This can be of use to trap a program that is running beyond control of the user in case he forgot to request Trace Mode, etc.

CHANGES TO DECUS NO. 8-150

When using DECUS No. 8-150 (PTOD8) with a PDP-8/S computer, a defect occurs which affects the 8/S. The problem is the combined micro instruction CMA RAL that occurs at location 1013 (low) and 7013 (high). This combination is illegal on the serial logic PDP-8/S. It is, however, valid on the PDP-8 machines with parallel internal logic.

Patch space is rather short in a neat program such as PTOD8, but the following patch will correct operation on PDP-8/S machines (PTOD8 low).

```
1013  5375            JMP PATCH 1
1175  7040  PATCH 1,  CMA
1176  7004            RAL
1177  5214            JMP 1014/RETURN
```

For PTOD8 (high) add 6000 to all addresses.

One other error has been found in the interpretive execution of ION and IOF instructions. At present, the pseudo-accumulator and pseudo-link are cleared erroneously by these instructions. The following patch rectifies this problem, but extends the program area to the end of the last program page.

(PTOD8 low) Add 6000 to addresses marked * for PTOD8 high.

```
*1402      5373            JMP PATCH 2
*1415      5373            JMP PATCH 2
*1417      5373            JMP PATCH 2
*1573      2777  PATCH 2,  ISZ PC
*1574      7000            NOP
*1575      5776            JMP TRRET
*1576    * 0602            TRRET
*1577    * 0753            PC
```

```
                /PTT TRACE AND OCTAL DEBUGGING PROGRAM  MODIFIED VERSION 1
                /FEATURES:REGISTER EXAMINATION AND MODIFICATION,
                /MULTIPLE BREAKPOINTS,MEMORY PROTECTION,INTERRUPT FACILITY
                /WORD SEARCH,TRACING A RUNNING PROGRAM,BINARY TAPE PUNCHING
                /PAGE 1

0200  6032  BEGIN,  6032            /INITIATE LOW SPEED READER
0201  6046          6046            /INITIATE LOW SPEED PUNCH
0202  3370          DCA TSW         /TSW:=0
0203  3365          DCA TMODS       /TMODS:=0
0204  5772          JMP I CLEAR     /CLEAR OTHER PROGRAM SWITCHES
0205  5336          JMP CLIST       /CLEAR LIST OF BREAKPOINTS
0206  4773  BEG2,   JMS I CRLF1     /PUNCH CRLF
0207  3371  BEG3,   DCA WORD       /CLEAR REGISTER WORD
0210  4774          JMS I READI     /READ NEXT CHARACTER
0211  4323          JMS SAD         /SKIP IF ACCU DIFFERS FROM NUMBER
0212  0215  CR,     215             /CARRIAGE RETURN?
0213  5250          JMP CRLF2       /YES
0214  4323          JMS SAD
0215  0212  LF,     212             /LINE FEED?
0216  5250          JMP CRLF2       /YES
0217  4323          JMS SAD
0220  0252          252             /ASTERISK?
0221  5775          JMP I STARI     /YES
0222  4776          JMS I PNCI      /FROM NOW ON PRINT CHAR.
0223  3354          DCA CSW         /CSW:=0
0224  6036          KRB             /READ SAME CHAR. FROM TTI BUFFER
0225  4323          JMS SAD
0226  0257  SLSH,   257             /SLASH?
0227  5242          JMP SLASH       /YES
0230  4323          JMS SAD
0231  0302          302             /LETTER B?
0232  5276          JMP BMOD        /YES
0233  4323          JMS SAD
0234  0324          324             /LETTER T?
0235  5311          JMP TMOD        /YES
0236  4323          JMS SAD
0237  0307          307             /LETTER G?
0240  5316          JMP GO          /YES
0241  5777          JMP I GOON      /NO, GO ON COMPARING
0242  1371  SLASH,  TAD WORD        /ADDRESS OF REGISTER TO BE EXAMINED
0243  3363          DCA REG         /STORE IN ORDER TO GO INDIRECT
0244  1763          TAD I REG       /CONTENT OF REQUIRED REGISTER
0245  4761          JMS I OCTP      /PUNCH IN OCTAL FORMAT
0246  3357          DCA LOCK        /LOCK:=0
0247  5207          JMP BEG3        /RETURN COMMAND TO OPERATOR
0250  1354  CRLF2,  TAD CSW         /SWITCH FOR CIPHER RECEPTION
0251  7650          SNA CLA         /FORE LAST CHAR. WAS A CIPHER?
0252  5256          JMP CLOSE       /NO
0253  1371          TAD WORD        /YES, DATA FOR REGISTER MOD.
0254  2357          ISZ LOCK        /(LOCK)=-1?
0255  3763          DCA I REG       /NO,MODIFY REGISTER
0256  7040  CLOSE,  CMA             /YES,NO MODIFICATION ALLOWED. ACCU:=
0257  3357          DCA LOCK        /(LOCK):=-1
0260  3354          DCA CSW         /CSW:=0,PREPARATION FOR NEXT NUMBER
0261  6036          KRB             /READ CHAR ONCE MORE FROM TTI BUFFER
0262  4323          JMS SAD
0263  0215          215             /CR?
0264  5206          JMP BEG2        /YES,PUNCH CRLF ,RETURN COMMAND
0265  7200          CLA
0266  2363          ISZ REG         /NO,EXAMINE NEXT REGISTER
0267  7000          NOP
0270  4773          JMS I CRLF1     /PUNCH CRLF
0271  1363          TAD REG         /ADDRESS NEXT REGISTER
0272  4761          JMS I OCTP      /PUNCH OCTAL
```

```
0273  1226  PSLH,   TAD SLSH          /SLASH
0274  4776          JMS I PNCI        /PUNCH SLASH
0275  5244          JMP SLASH+2       /RETURN TO EXAMINATIONROUTINE
0276  4344  BMOD,   JMS PREPR         /PREPARE POINTER AND COUNTER
0277  1762  BNXT,   TAD I PNTR
0300  7650          SNA CLA           /EMPTY PLACE IN LIST?
0301  5306          JMP.+5            /YES,INSERT BREAKPOINT
0302  2362          ISZ PNTR          /NO, TRY AGAIN
0303  2356          ISZ LCNT          /END OF LIST?
0304  5277          JMP BNXT          /NO
0305  5206          JMP BEG2          /YES RETURN COMMAND
0306  1371          TAD WORD          /REQUIRED BREAKADDRESS
0307  3762          DCA I PNTR        /INSERT IN LIST
0310  5206          JMP BEG2          /RETURN COMMAND TO OPERATOR
0311  1371  TMOD,   TAD WORD          /ADDRESS OF FIRST INSTR. TO BE TRACE
0312  3366          DCA TMREG         /STORE
0313  3370          DCA TSW           /CLEAR TRACE SWITCH
0314  2365          ISZ TMODS         /INDICATION THAT T MODE IS REQUIRED
0315  5206          JMP BEG2          /RETURN COMMAND
0316  3353  GO,     DCA LINK          /CLEAR LINK
0317  3352          DCA ACCU          /CLEAR ACCU
0320  1371          TAD WORD          /START ADDRESS OF PROGRAM TO BE HAND
0321  3364          DCA STADR
0322  5767          JMP I TRACER      /GO AND TRACE
0323  0000  SAD,    0                 /SUBROUTINE SKIP IF ACCU DIFFERS
0324  3344          DCA PREPR          /STORE CHAR TEMPORARELY
0325  1723          TAD I SAD         /NUMBER
0326  7041          CIA               / CONVERT INTO NEGATIVE
0327  1344          TAD PREPR         /ADD STORED NUMBER
0330  7650          SNA CLA           /EQUAL?
0331  5334          JMP.+3            /YES
0332  1344          TAD PREPR         /NO, RESTORE ACCU
0333  2323          ISZ SAD           /INCREMENT RETURN ADDRESS
0334  2323          ISZ SAD           /INCREMENT RETURN ADDRESS
0335  5723          JMP I SAD         /EXIT
0336  4344  CLIST,  JMS PREPR
0337  3762          DCA I PNTR        /CLEAR
0340  2362          ISZ PNTR          /STEP UP POINTER
0341  2356          ISZ LCNT          /READY?
0342  5337          JMP.-3            /NO,RETURN
0343  5206          JMP BEG2          /YES
0344  0000  PREPR,  0                 /SUBROUTINE TO INIT. POINTER AND LCN
0345  1360          TAD M40           /LENGTH OF LIST
0346  3356          DCA LCNT
0347  1355          TAD LADR          /ADDRESS OF LIST
0350  3362          DCA PNTR
0351  5744          JMP I PREPR       /EXIT


              /CONSTANTS AND INDIRECT ADDRESSES PAGE 1
0352  0000  ACCU,   0
0353  0000  LINK,   0
0354  0000  CSW,    0
0355  1525  LADR,   BLST
0356  0000  LCNT,   0
0357  0000  LOCK,   0
0360  7740  M40,    -40
0361  0671  OCTP,   OCPN
0362  0000  PNTR,   0
0363  0000  REG,    0
0364  0000  STADR,  0
0365  0000  TMODS,  0
0366  0000  TMREG,  0
0367  0600  TRACER, TRACE
0370  0000  TSW,    0
0371  0000  WORD,   0
0372  1227  CLEAR,  CLR
```

```
0373  1221  CRLF1,  PCL
0374  0745  READI,  READ
0375  0535  STARI,  STAR
0376  0726  PNCI,   PUNCH
0377  0400  GOON,   PROC
            *BEGIN+200
0400  4747  PROC,   JMS I SAD1      /INVESTIGATION OF COMMAND SIGNS
0401  0305          305             /LETTER E?
0402  5751          JMP I ERSE      /YES ERASE ALL BREAKPOINTS
0403  4747          JMS I SAD1
0404  0322          322             /LETTER R?
0405  5231          JMP RMUV        /YES REMOVE 1 BREAKPOINT
0406  4747          JMS I SAD1      /NO
0407  0301          301             /LETTER A?
0410  5246          JMP ACU         /YES
0411  4747          JMS I SAD1      /NO
0412  0314          314             /LETTER L?
0413  5245          JMP LNK         /YES
0414  4747          JMS I SAD1
0415  0303          303             /LETTER C?
0416  5750          JMP I INTP1     /GO TO INTERPRETATOR
0417  4747          JMS I SAD1
0420  0315          315             /LETTER M?
0421  5253          JMP MPRT        /YES MEMORY PROTECTION MODE
0422  4747          JMS I SAD1
0423  0320          320             /LETTER P?
0424  5264          JMP BINT        /YES,PUNCH TAPE IN BINARY
0425  4747          JMS I SAD1
0426  0273          273             /;?
0427  5261          JMP FADR        /YES , FIRST ADDRESS OF BLOCK
0430  5775          JMP I GOON1     /INVESTIGATION OF COMMAND SIGNS
0431  4752  RMUV,   JMS I PREP1     /PREPARE LIST HANDLING
0432  1753          TAD I PNTR1
0433  3346          DCA TEM3        /STORE TO GO INDIRECT
0434  1746  RNXT,   TAD I TEM3      /CONTENT OF FIRST PLACE IN LIST
0435  7041          CIA             /CONVERT TO NEGATIVE
0436  1754          TAD I WRDI      /ADD ADDRESS OF BREAKPOINT
0437  7650          SNA CLA         /EQUAL?
0440  3746          DCA I TEM3      /YES, REMOVE
0441  2346          ISZ TEM3        /NO, TRY NEXT
0442  2755          ISZ I LCNT1     /END OF LIST?
0443  5234          JMP RNXT        /NO
0444  5756          JMP I BEG2I     /YES RETURN COMMAND
0445  7201  LNK,    CLA IAC         /ACCU:=1, ADDRESS LINK=ADDRESS ACCU+
0446  1357  ACU,    TAD ACCUI       /ADDRESS OF PROGRAM ACCU
0447  3760          DCA I REG1      /FILL REG FOR REG EXAMINATION
0450  1361          TAD SP          /SPACE
0451  4763          JMS I PNCI1     /PUNCH
0452  5762          JMP I PSLH1     /PUNCH SLASH
0453  2365  MPRT,   ISZ MSW         /MEMORY PROTECT SWITCH:=1
0454  1754          TAD I WRDI      /LAST ADDRESS OF BLOCK
0455  3364          DCA EADR        /STORE
0456  1343          TAD TBADR       /TEMP. FIRST ADDRESS OF BLOCK
0457  3367          DCA BADR        /STORE IN FIRST ADDR. OF MEMPROT.
0460  5756          JMP I BEG2I     /RETURN COMMAND
0461  1754  FADR,   TAD I WRDI      /FIRST ADDRESS OF BLOCK
0462  3343          DCA TBADR       /STORE
0463  5766          JMP I BEG3I     /RETURN COMMAND WITHOUT CRLF
0464  7402  BINT,   HLT             /ROUTINE TO PUNCH BINARY TAPE
0465  4776          JMS I PTRI      /ROUTINE TRAILER-LEADER PUNCH
0466  4745          JMS I RSTI      /RESTORE PLACE 1 AND 2
0467  1754          TAD I WRDI      /LAST ADDRESS OF BLOCK TO BE PUNCHED
```

```
0470  3364          DCA EADR          /STORE
0471  1370          TAD C100          /CHARACTER
0472  3371          DCA ASRS          /IDENTIFICATION OF FIRST ADDRESS
0473  1364          TAD EADR
0474  7040          CMA
0475  1343          TAD TBADR         /NUMBER OF WORDS TO PUNCH
0476  3372          DCA BCNT          /STORE
0477  1343          TAD TBADR         /FIRST AND CONSEQUENT ADDRESSES
0500  4311          JMS PBIN          /PUNCH BINARY
0501  3371          DCA ASRS          /CLEAR IDENTIFICATION MARK
0502  1743          TAD I TBADR       /INSTRUCTION
0503  4311          JMS PBIN          /PUNCH BINARY
0504  2343          ISZ TBADR         /STEP UP POINTER OF BLOCK
0505  7000          NOP               /IN CASE ISZ SKIPS
0506  2372          ISZ BCNT          /END OF BLOCK?
0507  5302          JMP.-5            /NO, NEXT WORD
0510  5766          JMP I BEG3I       /RETURN COMMAND WITHOUT CRLF
0511  0000  PBIN,    0                 /PUNCH ONE WORD BINARY
0512  3346          DCA TEM3          /STORE WORD
0513  1346          TAD TEM3
0514  7012          RTR
0515  7012          RTR
0516  7012          RTR               /MOST SIGNIFICANT PART FIRST
0517  0373          AND M77           /MASK 6 BITS
0520  1371          TAD ASRS          /ADD IDENTIFICATION MARK
0521  4326          JMS PCHS          /PUNCH AND KEEP CHECKSUM
0522  1346          TAD TEM3
0523  0373          AND M77           /LEAST SIGNIFICANT PART OF THE WORD
0524  4326          JMS PCHS
0525  5711          JMP I PBIN        /EXIT
0526  0000  PCHS,    0
0527  6046          6046
0530  6041          6041
0531  5330          JMP.-1
0532  1374          TAD CHS           /CHECKSUM
0533  3374          DCA CHS           /ADD UP
0534  5726          JMP I PCHS        /EXIT
0535  3777  STAR,    DCA I LFLGI
0536  1374          TAD CHS           /PUNCH CHECKSUM, END  BINARY TAPE
0537  4311          JMS PBIN
0540  3374          DCA CHS           /CLEAR CHECKSUM
0541  4744          JMS I PTRALI
0542  5766          JMP I BEG3I       /RETURN COMMAND WITHOUT CRLF
            /CONSTANTS AND VARIABLES TRACER PAGE 2
0543  0000  TBADR,  0
0544  1430  PTRALI, PTRAIL
0545  1420  RSTI,   RST
0546  0000  TEM3,   0
0547  0323  SAD1,   SAD
0550  1000  INTP1,  INTP
0551  0336  ERSE,   CLIST
0552  0344  PREP1,  PREPR
0553  0362  PNTR1,  PNTR
0554  0371  WRDI,   WORD
0555  0356  LCNT1,  LCNT
0556  0206  BEG2I,  BEG2
0557  0352  ACCUI,  ACCU
0560  0363  REG1,   REG
0561  0240  SP,     240
0562  0273  PSLH1,  PSLH
0563  0726  PNCI1,  PUNCH
0564  0000  EADR,   0
0565  0000  MSW,    0
0566  0207  BEG3I,  BEG3
0567  0000  BADR,   0
```

```
0570  0100  C100,   '100'
0571  0000  ASRS,   0
0572  0000  BCNT,   0
0573  0077  M77,    77
0574  0000  CHS,    0
0575  1463  GOON1,  PROC2
0576  1440  PTRI,   LEAD
0577  1500  LFLGI,  LFLAG
            *BEGIN+400

0600  1752  TRACE,  TAD I STAD1     /ROUTINE TO DETERMINE WHETHER NEXT
0601  353           DCA PC          /INSTRUCTION HAS TO BE PRINTED OR NOT
0602  1354  TRRET,  TAD IONF        /PROGRAM INTERRUPT SWITCH
0603  7640          SZA CLA         /HAS THERE BEEN ION?
0604  6001          ION             /YES ENABLE INTERRUPTS
0605  1757          TAD I TMOD1     /TRACE MODE REQUESTED ?
0606  7650          SNA CLA         /NO
0607  5226          JMP SBPT        /NO, IS IT BREAKPOINT?
0610  6002          IOF             /YES , DISABLE INTERRUPTS
0611  1760          TAD I TSW1
0612  7640          SZA CLA         /IS TRACING ACTIVE ALREADY?
0613  5243          JMP PINS        /YES, PRINT
0614  1761          TAD I TMRG1     /NO, FIRST ADDRESS TO BE TRACED
0615  7100          CLL
0616  7041          CIA
0617  1353          TAD PC          /PC - TRACE ADDRESS
0620  4316          JMS COMPR       /COMPARE ROUTINE
0621  7410          SKP             /EQUAL
0622  5226          JMP SBPT        /PC<TMREG,IS IT BREAKPOINT?
0623  7001          IAC             /PC>TMREG
0624  3760          DCA I TSW1      /TSW:=1
0625  5243          JMP PINS        /PRINT INSTRUCTION,ACCU ETC
0626  4755  SBPT,   JMS I PREPI
0627  1762          TAD I PNTR2
0630  3253          DCA PRIN        /STORE TO GO INDIRECT
0631  1653  SNXT,   TAD I PRIN
0632  7041          CIA
0633  1353          TAD PC          /COMPARE WORDS IN BREAKLIST WITH PC
0634  7650          SNA CLA         /EQUAL?
0635  5243          JMP PINS        /YES PRINT
0636  2253          ISZ PRIN        /STEP UP POINTER
0637  2763          ISZ I LCNT2     /END OF LIST?
0640  5231          JMP SNXT        /NO, NEXT
0641  6002          IOF             /DISABLE INTERRUPTS
0642  5764          JMP I INTP2     /EXIT
0643  6002  PINS,   IOF             /DISABLE INTERRUPTS
0644  4253          JMS PRIN        /PRINT ADDRESS, LINK, ACCU AND INSTR
0645  7604          LAS             /LOAD WITH SWITCHREG
0646  7650          SNA CLA
0647  5764          JMP I INTP2     /CONTINUE
0650  1353  PRET,   TAD PC
0651  3765          DCA I REG2      /LOAD REG WITH PC FOR
0652  5756          JMP I SLS1      /REGISTER EXAMINATION
0653  0000  PRIN,   0
0654  4766          JMS I CRLF3     /CRLF
0655  1353          TAD PC          /PROGRAM COUNTER
0656  4271          JMS OCPN        /PRINT OCTAL
0657  1767          TAD I LINK1     /LINK
0660  1372          TAD C260        /CHARACTER IN ASCII
0661  4326          JMS PUNCH
0662  1371          TAD SP1         /SPACE
0663  4326          JMS PUNCH
0664  1770          TAD I ACCU1     /ACCU
0665  4271          JMS OCPN
```

```
0666  1753          TAD I PC          /INSTRUCTION
0667  4271          JMS OCPN
0670  5653          JMP I PRIN        /EXIT
0671  0000  OCPN,    0                 /SUBROUTINE PRINT NUMBER OCTAL
0672  3316          DCA COMPR
0673  1374          TAD M4            /-4
0674  3375          DCA OCNT          /INITIATE COUNTER
0675  1316          TAD COMPR         /NUMBER
0676  7004          RAL
0677  7004          RAL
0700  7006          RTL               /SHIFT MOST SIGNIFICANT OCTAL
0701  3316          DCA COMPR         /STORE
0702  1316          TAD COMPR
0703  0373          AND MS7           /MASK 3 BITS
0704  1372          TAD C260          /CHARACTER IN ASCII
0705  4326          JMS PUNCH
0706  1316          TAD COMPR
0707  2375          ISZ OCNT          /READY?
0710  5277          JMP.-11           /NO, NEXT OCTAL
0711  7300          CLA CLL           /YES
0712  3316          DCA COMPR
0713  1371          TAD SP1           /SPACE
0714  4326          JMS PUNCH
0715  5671          JMP I OCPN        /EXIT
0716  0000  COMPR,   0                 /SUBROUTINE TO COMPARE TWO NUMBERS
0717  7650          SNA CLA           /DIFFERENCE IN ACCU
0720  5716          JMP I COMPR       /EQUAL
0721  7430          SZL
0722  2316          ISZ COMPR         />
0723  2316          ISZ COMPR         /<
0724  7100          CLL
0725  5716          JMP I COMPR       /EXIT
0726  0000  PUNCH,   0
0727  6046          6046
0730  6041          6041
0731  5330          JMP.-1
0732  7200          CLA
0733  5726          JMP I PUNCH
0734  7200  INTD,    CLA
0735  3354          DCA IONF          /IONF:=0 UNTIL NEXT ION
0736  3777          DCA I IONFHI      /CLEAR IONFH
0737  1353          TAD PC
0740  3000          DCA 0             /RETURN ADDRESS TO 0
0741  4776          JMS I RESRE
0742  7001          IAC               /ACCU:=1
0743  3353          DCA PC            /SET PC TO 1
0744  5202          JMP TRRET         /TRACER
0745  0000  READ,    0
0746  6031          6031
0747  5346          JMP.-1
0750  6036          6036
0751  5745          JMP I READ
              /CONSTANTS AND VARIABLES TRACER PAGE3
0752  0364  STAD1,   STADR
0753  0000  PC,      0
0754  0000  IONF,    0
0755  0344  PREPI,   PREPR
0756  0246  SLS1,    SLASH+4
0757  0365  TMOD1,   TMODS
0760  0370  TSW1,    TSW
0761  0366  TMRG1,   TMREG
0762  0362  PNTR2,   PNTR
0763  0356  LCNT2,   LCNT
0764  1000  INTP2,   INTP
```

```
0765  0363  REG2,   REG
0766  1221  CRLF3,  PCL
0767  0353  LINK1,  LINK
0770  0352  ACCU1,  ACCU
0771  0240  SP1,    240
0772  0260  C260,   260
0773  0007  MS7,    7
0774  7774  M4,     -4
0775  0000  OCNT,   0
0776  1044  RESRE,  REST
0777  1172  IONFHI, IONFH

            *BEGIN+600
1000  1372  INTP,   TAD IONFH
1001  3750          DCA I IONF1
1002  1740          TAD I PCI       /INTERPRETATION ROUTINE
1003  3341          DCA PCD         /STORE TO GO INDIRECT
1004  1742          TAD I MSWI      /MEMORY PROTECT SWITCH
1005  7650          SNA CLA         /PROTECTION REQUESTED?
1006  5212          JMP.+4          /NO
1007  1341          TAD PCD         /PROGRAM COUNTER
1010  4764          JMS I INBL1     /IN PROT. BLOCK?
1011  5252          JMP PMES        /YES, PRINT MESSAGE
1012  1741          TAD I PCD       /NO
1013  7044          CMA RAL         /IF LINK=0 AND ACCU POS. THEN
1014  7730          SPA SZL CLA     /GROUP 6 OR 7
1015  5260          JMP GR5         /NO, GROUP 5
1016  1741          TAD I PCD       /NO,INSTRUCTION
1017  1343          TAD MION        /-6001
1020  7450          SNA             /IS IT ION?
1021  5771          JMP I PIONI     /YES
1022  1344          TAD M1          /,NO, ADD -1
1023  7650          SNA CLA         /IS IT IOF?
1024  5770          JMP I PIOFI     /YES
1025  1741          TAD I PCD       /INSTRUCTION
1026  3232  DO,     DCA .+4         /STORE INSTRUCTION
1027  1745          TAD I LINK2     /LINK
1030  7110          CLL RAR         /RESTORE LINK
1031  1746          TAD I ACCU2     /ACCU
1032  0000          0
1033  7410          SKP             /NON SKIPPING INSTRUCTION
1034  2740          ISZ I PCI       /STEP UP PROGRAM COUNTER
1035  7000          NOP
1036  2740          ISZ I PCI       /SKIPPING INSTRUCTION
1037  7000          NOP
1040  3746          DCA I ACCU2     /SAVE ACCU
1041  7204          GLK
1042  3745          DCA I LINK2     /SAVE LINK
1043  5747          JMP I TRAC1     /GO TO TRACING PROGRAM
1044  0000  REST,   0               /RESTORE 1 AND 2
1045  1373          TAD DUM1
1046  3001          DCA 1
1047  1374          TAD DUM2
1050  3002          DCA 2
1051  5644          JMP I REST
1052  4751  PMES,   JMS I PINS1     /PRINT INSTRUCTION
1053  1352          TAD F           /LETTER F
1054  4754          JMS I PRNT      /PRINT
1055  1353          TAD A           /LETTER A
1056  4754          JMS I PRNT
1057  5755          JMP I PRETI     /RETURN COMMAND
1060  1741  GR5,    TAD I PCD       /INSTRUCTION
1061  0356          AND MS177       /MASK RELATIVE ADDRESS BITS
1062  3357          DCA TEM1        /STORE TEMPORARELY
1063  1741          TAD I PCD
```

```
1064  7006          RTL
1065  7006          RTL                 /I-BIT TO LINK, P-BIT TO SIGN
1066  7700          SMA CLA             /IS P-BIT PRESENT/
1067  5327          JMP AUTO            /NO,PAGE 0 ,LOOK FOR AUTOINDEX
1070  1341          TAD PCD             /YES
1071  0360          AND MS7600          /PAGE BITS OF ADDRESS
1072  1357          TAD TEM1
1073  3357          DCA TEM1            /ABSOLUTE ADDRESS OF MEMORY
1074  7420  ADID,   SNL                 /REFERENCE INSTRUCTION
1075  5300          JMP.+3              /NO I-BIT
1076  1757          TAD I TEM1          /I-BIT, GO INDIRECT
1077  3357          DCA TEM1
1100  1742  MPRT1,  TAD I MSWI          /MEMORY PROTECTION ABSOLUTE ADDRESS
1101  7650          SNA CLA             /REQUESTED?
1102  5306          JMP.+4              /NO, FORGET ABOUT IT
1103  1357          TAD TEM1            /YES, ABSOLUTE ADDRESS
1104  4764          JMS I INBL1         /IN PROTECTED BLOCK?
1105  5252          JMP PMES            /YES, PRINT MESSAGE
1106  1741          TAD I PCD           /NO, INSTRUCTION
1107  0367          AND MS7000          /OP CODE
1110  1361          TAD M4000           /-4000
1111  7450          SNA                 /IS IT JMS?
1112  5320          JMP PJMS            /YES
1113  1362          TAD M1000           /IS IT JMP?
1114  7450          SNA
1115  5324          JMP PJMP            /YES
1116  1363          TAD IST             /IST=JMP I TEM1
1117  5226          JMP DO
1120  7001  PJMS,   IAC                 /RETURN ADDRESS SUBROUTINE
1121  1740          TAD I PCI           /+1
1122  3757          DCA I TEM1          /STORE IN FIRST PLACE
1123  7001          IAC                 /OF SUBROUTINE
1124  1357  PJMP,   TAD TEM1            /ADDRESS NEXT INSTRUCTION
1125  3740          DCA I PCI           /TO BE EXECUTED
1126  5747          JMP I TRAC1         /GO TO TRACING PROGRAM
1127  7420  AUTO,   SNL                 /I-BIT?
1130  5300          JMP MPRT1           /NO,RETURN
1131  1741          TAD I PCD           /INSTRUCTION
1132  0365          AND MS370           /MASK TO FIND AUTOINDEX
1133  1366          TAD M10
1134  7640          SZA CLA             /AUTOINDEX?
1135  5276          JMP ADID+2          /NO
1136  2757          ISZ I TEM1          /YES,INCREMENT CONTENTS OF TEM1
1137  5276          JMP ADID+2          /RETURN
              /CONSTANTS AND VARIABLES TRACER PAGE4
1140  0753  PCI,    PC
1141  0000  PCD,    0
1142  0565  MSWI,   MSW
1143  1777  MION,   -6001
1144  7777  M1,     -1
1145  0353  LINK2,  LINK
1146  0352  ACCU2,  ACCU
1147  0602  TRAC1,  TRRET
1150  0754  IONF1,  IONF
1151  0653  PINS1,  PRIN
1152  0306  F,      306
1153  0301  A,      301
1154  0726  PRNT,   PUNCH
1155  0650  PRETI,  PRET
1156  0177  MS177,  177
1157  0000  TEM1,   0
1160  7600  MS7600, 7600
1161  4000  M4000,  4000
1162  7000  M1000,  -1000
1163  5757  IST,    JMP I TEM1
```

```
1164  1200  INBL1,  INBLK
1165  0370  MS370,  370
1166  7770  M10,    -10
1167  7000  MS7000, 7000
1170  1416  PIOFI,  PIOF
1171  1400  PIONI,  PION
1172  0000  IONFH,  0
1173  0000  DUM1,   0
1174  0000  DUM2,   0
            *BEGIN+1000

1200  0000  INBLK,  0              /SUBROUTINE USED FOR MEMORY
1201  7100        CLL
1202  3221        DCA PCL          /PROTECTION
1203  1744        TAD I BADR1
1204  7041        CIA
1205  1221        TAD PCL          /DIFFERENCE
1206  4745        JMS I COM1       /COMPARE ROUTINE
1207  5600        JMP I INBLK      /EQUAL, FIRST EXIT MEANS
1210  5217        JMP OUTB         /IN BLOCK, SECOND MEANS OUT BLOCK
1211  1746        TAD I EADR1
1212  7041        CIA
1213  1221        TAD PCL          /DIFFERENCE WITH UPPER LIMIT
1214  4745        JMS I COM1
1215  5600        JMP I INBLK      /FIRST EXIT MEANS IN BLOCK
1216  5600        JMP I INBLK      /SECOND ALSO
1217  2200  OUTB, ISZ INBLK       /SECOND
1220  5600        JMP I INBLK      /OUT OF BLOCK
1221  0000  PCL,  0                /SUBROUTINE TO PUNCH CRLF
1222  1370        TAD CRT          /CR
1223  4765        JMS I PNCI2      /PUNCH
1224  1371        TAD LFD          /LF
1225  4765        JMS I PNCI2
1226  5621        JMP I PCL        /EXIT
1227  4740  CLR,  JMS I PRCL       /CLEAR MSW AND RESTORE 1,2
1230  3747        DCA I IONF2      /CLEAR IONFLAG
1231  3750        DCA I CHS2       /CLEAR CHECKSUM
1232  3751        DCA I TSW2       /CLEAR TRACE SWITCH
1233  3752        DCA I TMD2       /CLEAR TRACE MODE SWITCH
1234  3761        DCA I CSWI       /CLEAR CIPHER SWITCH
1235  3742        DCA I IONFH2
1236  5753        JMP I CLST1      /RETURN
1237  4774  PROC1, JMS I SAD2      /CONTINUE COMMAND RECOGNITION
1240  0246        246              /& FOR MASK?
1241  5277        JMP PAND         /YES
1242  4774        JMS I SAD2
1243  0332        332              /LETTER Z?
1244  5302        JMP PSERC        /YES
1245  4774        JMS I SAD2
1246  0323        323              /LETTER S?
1247  5336        JMP SKIP         /YES
1250  4774        JMS I SAD2
1251  0337        337              /CORRECTION SIGN?
1252  5760        JMP I BEG3I1     /YES SKIP READ WORD
1253  3366        DCA TEM2         /STORE CHARACTER FOR CIPHER
1254  1366        TAD TEM2         /RECOGNITION
1255  0372        AND MAS370
1256  1373        TAD M260         / STRIP AND SUBTRACT
1257  7640        SZA CLA          /IS IT 0-7?
1260  5274        JMP QMRK         /NO, PRINT?
1261  1366        TAD TEM2         /YES
1262  0343        AND MASK7        /MASK CIPHER BITS
1263  3366        DCA TEM2         /STORE
1264  7001        IAC
1265  3761        DCA I CSWI       /CSW:=1
```

```
1266  1762          TAD I WRD2      /WORD
1267  7106          CLL RTL
1270  7004          RAL             /FORM OCTAL NUMBER
1271  1366          TAD TEM2        /ADD NEW CIPHER
1272  3762          DCA I WRD2      /RESTORE WORD
1273  5763          JMP I BEGN      /RETURN COMMAND
1274  1364  QMRK,   TAD C277        /QUESTIONMARK
1275  4765          JMS I PNCI2     /PUNCH
1276  5760          JMP I BEG3I1    /RETURN COMMAND
1277  1762  PAND,   TAD I WRD2      /WORD
1300  3355          DCA MASK        /STORE MASK FOR WORD SEARCH
1301  5760          JMP I BEG3I1    /RETURN COMMAND
1302  1775  PSERC,  TAD I WBADRI
1303  3366          DCA TEM2        /STORE TO GO INDIRECT
1304  1766  SRNXT,  TAD I TEM2      /WORD ON THIS ADDRESS
1305  0355          AND MASK
1306  7041          CIA
1307  1762          TAD I WRD2      /COMPARE
1310  7650          SNA CLA         /EQUAL?
1311  5326          JMP FND         /YES
1312  1366  TEND,   TAD TEM2        /NO
1313  7041          CIA
1314  1776          TAD I WEADRI    /END OF BLOCK?
1315  7640          SZA CLA
1316  5322          JMP.+4          /NO
1317  7240          STA             /YES
1320  3355          DCA MASK        /MASK:=7777
1321  5767          JMP I BEG2I2    /RETURN COMMAND
1322  7001          IAC             /ACCU:=1
1323  1366          TAD TEM2
1324  3366          DCA TEM2        /STEP UP BEGA
1325  5304          JMP SRNXT       /TRY NEXT
1326  4221  FND,    JMS PCL         /PRINT CRLF
1327  1366          TAD TEM2        /ADDRESS
1330  4754          JMS I OCTPR     /PRINT OCTAL
1331  1341          TAD C334        /BACKSLASH
1332  4765          JMS I PNCI2     /PUNCH
1333  1766          TAD I TEM2      /PUNCH CONTENTS
1334  4754          JMS I OCTPR
1335  5312          JMP TEND        /FIND OTHER WORDS
1336  1356  SKIP,   TAD SKIPIN      /7410
1337  5757          JMP I DOI       /EXECUTE AN UNCONDITIONAL SKIP
      /CONSTANTS AND VARIABLES TRACER PAGE5
1340  1454  PRCL,   CLCON
1341  0334  C334,   334
1342  1172  IONFH2, IONFH
1343  0007  MASK7,  7
1344  0567  BADR1,  BADR
1345  0716  COM1,   COMPR
1346  0564  EADR1,  EADR
1347  0754  IONF2,  IONF
1350  0574  CHS2,   CHS
1351  0370  TSW2,   TSW
1352  0365  TMD2,   TMODS
1353  0336  CLST1,  CLIST
1354  0671  OCTPR,  OCPN
1355  7777  MASK,   7777
1356  7410  SKIPIN, 7410
1357  1026  DOI,    DO
1360  0207  BEG3I1, BEG3
1361  0354  CSWI,   CSW
1362  0371  WRD2,   WORD
1363  0210  BEGN,   BEG3+1
1364  0277  C277,   277
1365  0726  PNCI2,  PUNCH
```

```
1366  0000  TEM2,    0
1367  0206  BEG2I2,  BEG2
1370  0215  CRT,     215
1371  0212  LFD,     212
1372  0370  MAS370,  370
1373  7520  M260,    -260
1374  0323  SAD2,    SAD
1375  1521  WBADRI,  WBADR
1376  1520  WEADRI,  WEADR
            *BEGIN+1200
1400  1701  PION,    TAD I IONFH3
1401  7640           SZA CLA
1402  5702           JMP I DO7I        /NOT FIRST ION MET
1403  1001           TAD 1
1404  3703           DCA I DUM1I       /SAVE LOC. 1
1405  1002           TAD 2
1406  3704           DCA I DUM2I       /SAVE LOC. 2
1407  1305           TAD INIS          /TRAP
1410  3001           DCA 1
1411  1306           TAD INTDI
1412  3002           DCA 2
1413  7001           IAC
1414  3701           DCA I IONFH3      /IONFH:=1
1415  5702           JMP I DO7I
1416  4220  PIOF,    JMS RST           /RESTORE LOC 1 AND 2
1417  5702           JMP I DO7I        /RETURN
1420  1420  RST,     0                 /RESTRE ROUTINE
1421  1701           TAD I IONFH3
1422  7650           SNA CLA           /IONFH3=1?
1423  7410           SKP               /NO
1424  4710           JMS I RESTI       /RESTORE
1425  3701           DCA I IONFH3
1426  3707           DCA I IONACT      /CLEAR ION FLAGS
1427  5620           JMP I RST
1430  1420  PTRAIL,  0
1431  1311           TAD M200          /ROUTINE TRAIL PUNCH
1432  3312           DCA COUNT
1433  1313           TAD C200
1434  4714           JMS I PNCI3       /PUNCH TRAIL
1435  2312           ISZ COUNT         /READY?
1436  5233           JMP.-3            /NO
1437  5630           JMP I PTRAIL      /YES
1440  0000  LEAD,    0                 /ROUTINE LEADER PRINT?
1441  1300           TAD LFLAG         /FIRST TIME BINPUNCH?
1442  7650           SNA CLA
1443  4230           JMS PTRAIL        /YES, PUNCH TRAIL
1444  2300           ISZ LFLAG         /LFLAG
1445  5640           JMP I LEAD        /EXIT
1446  0000  CATCH,   0                 /CATCH DEPENDS ON SR
1447  7604           LAS
1450  7710           SPA CLA           /SR NEGATIVE?
1451  5676           JMP I PRINT       /YES, PRINT INSTR.
1452  1677           TAD I TMOD2       /NO
1453  5646           JMP I CATCH       /RETURN TO TRACER
1454  0000  CLCON,   0                 /RESTORE BEFORE RESTART PTOD
1455  4220           JMS RST
1456  7240           STA
1457  3675           DCA I MSK         /SET WORD MASK TO 7777
1460  3674           DCA I MSWI1       /CLEAR MSW
1461  3724           DCA I LFLI        /CLEAR LEADER FLAG
1462  5654           JMP I CLCON       /CONTINUE CLEARING
1463  4715  PROC2,   JMS I SAD3       /INVESTIGATE COMMANDS
1464  0327           327               /LETTER W?
1465  5267           JMP WBLK          /YES
1466  5716           JMP I GOON2       /NO, CONTINUE
```

```
1467  1717  WBLK,   TAD I WRD3      /LAST WORD TYPED BY USER
1470  3320          DCA WEADR       /STORE
1471  1722          TAD I TBADRI    /FIRST WORD TYPED BY USER
1472  3321          DCA WBADR       /STORE
1473  5723          JMP I BEG2I1    /RETURN COMMAND WITH CRLF
```

```
                    *TRRET+3

0605  4757          JMS I TRACE+157

                    *TRACE+157
0757  1446          CATCH
```

```
                    /CONSTANTS AND VARIABLES PAGE6
1474  0565  MSWI1,  MSW
1475  1355  MSK,    MASK
1476  0643  PRINT,  PINS
1477  0365  TMOD2,  TMODS
1500  0000  LFLAG,  0
1501  1172  IONFH3, IONFH
1502  1035  D07I,   DO+7
1503  1173  DUM1I,  DUM1
1504  1174  DUM2I,  DUM2
1505  5402  INIS,   JMP I 2
1506  0734  INTDI,  INTD
1507  0754  IONACT, IONF
1510  1044  RESTI,  REST
1511  7600  M200,   -200
1512  0000  COUNT,  0
1513  0200  C200,   200
1514  0726  PNCI3,  PUNCH
1515  0323  SAD3,   SAD
1516  1237  GOON2,  PROC1
1517  0371  WRD3,   WORD
1520  0000  WEADR,  0
1521  0000  WBADR,  0
1522  0543  TBADRI, TBADR
1523  0206  BEG2I1, BEG2
1524  1500  LFLI,   LFLAG
1525  0000  BLST,   0
```

A	1153
ACCU	0352
ACCUI	0557
ACCU1	0770
ACCU2	1146
ACU	0446
ADID	1074
ASRS	0571
AUTO	1127
BADR	0567
BADR1	1344
BCNT	0572
BEGIN	0200
BEGN	1363
BEG2	0206
BEG2I	0556
BEG2I1	1523
BEG2I2	1367
BEG3	0207
BEG3I	0566
BEG3I1	1360
BINT	0464
BLST	1525
BMOD	0276
BNXT	0277
CATCH	1446
CHS	0574
CHS2	1350
CLCON	1454
CLEAR	0372
CLIST	0336
CLOSE	0256
CLR	1227
CLST1	1353
COMPR	0716
COM1	1345
COUNT	1512
CR	0212
CRLF1	0373
CRLF2	0250
CRLF3	0766
CRT	1370
CSW	0354
CSWI	1361
C100	0570
C200	1513
C260	0772
C277	1364
C334	1341
DO	1026
DOI	1357
DO7I	1502
DUM1	1173
DUM1I	1503
DUM2	1174
DUM2I	1504
EADR	0564
EADR1	1346
ERSE	0551
F	1152
FADR	0461
FND	1326
GO	0316
GOON	0377
GOON1	0575
GOON2	1516
GR5	1060
INBLK	1200
INBL1	1164
INIS	1505
INTD	0734
INTDI	1506
INTP	1000
INTP1	0550
INTP2	0764
IONACT	1507
IONF	0754
IONFH	1172
IONFHI	0777
IONFH2	1342
IONFH3	1501
IONF1	1150
IONF2	1347
IST	1163
LADR	0355
LCNT	0356
LCNT1	0555
LCNT2	0763
LEAD	1440
LF	0215
LFD	1371
LFLAG	1500
LFLGI	0577
LFLI	1524
LINK	0353
LINK1	0767
LINK2	1145
LNK	0445
LOCK	0357
MASK	1355
MASK7	1343
MAS370	1372
MION	1143
MPRT	0453
MPRT1	1100
MSK	1475
MSW	0565
MSWI	1142
MSWI1	1474
MS177	1156
MS370	1165
MS7	0773
MS7000	1167
MS7600	1160
M1	1144
M10	1166
M1000	1162
M200	1511
M260	1373
M4	0774
M40	0360
M4000	1161
M77	0573
OCNT	0775
OCPN	0671
OCTP	0361
OCTPR	1354
OUTB	1217
PAND	1277
PBIN	0511
PC	0753
PCD	1141
PCHS	0526
PCI	1140
PCL	1221
PINS	0643
PINS1	1151
PIOF	1416
PIOFI	1170
PION	1400
PIONI	1171
PJMP	1124
PJMS	1120
PMES	1052
PNCI	0376
PNCI1	0563
PNCI2	1365
PNCI3	1514
PNTR	0362
PNTR1	0553
PNTR2	0762
PRCL	1340
PREPI	0755
PREPR	0344
PREP1	0552
PRET	0650
PRETI	1155
PRIN	0653
PRINT	1476
PRNT	1154
PROC	0400
PROC1	1237
PROC2	1463
PSERC	1302
PSLH	0273
PSLH1	0562
PTRAIL	1430
PTRALI	0544
PTRI	0576
PUNCH	0726
QMRK	1274
READ	0745
READI	0374
REG	0363
REG1	0560
REG2	0765
RESRE	0776
REST	1044
RESTI	1510
RMUV	0431
RNXT	0434
RST	1420
RSTI	0545
SAD	0323
SAD1	0547
SAD2	1374
SAD3	1515
SBPT	0626
SKIP	1336
SKIPIN	1356
SLASH	0242
SLSH	0226
SLS1	0756
SNXT	0631
SP	0561
SP1	0771
SRNXT	1304
STADR	0364
STAD1	0752
STAR	0535
STARI	0375
TBADR	0543
TBADRI	1522
TEM1	1157
TEM2	1366
TEM3	0546
TEND	1312
TMD2	1352
TMOD	0311
TMODS	0365
TMOD1	0757
TMOD2	1477
TMREG	0366
TMRG1	0761
TRACE	0600
TRACER	0367
TRAC1	1147
TRRET	0602
TSW	0370
TSW1	0760
TSW2	1351
WBADR	1521
WBADRI	1375
WBLK	1467
WEADR	1520
WEADRI	1376
WORD	0371
WRDI	0554
WRD2	1362
WRD3	1517